

Rust⁻: A Simple Programming Language

Programming Assignment 3

Code Generation

Due Date: 1:20PM, Tuesday, June 12, 2018

Your assignment is to generate code (in Java assembly language) for the *Rust*⁻ language. The generated code will then be translated to Java bytecode by a Java assembler.

1 Assignment

Your assignment will be divided into the following parts:

- parsing declarations for constants and variables
- code generation for expressions and statements
- code generation for conditional statements and while loops
- code generation for procedure calls

1.1 Language Restrictions

In order to keep the code generation assignment simple that we can implement most of the features of the language, only a subset set of *Rust*⁻ language will be considered in this assignment:

- No READ statements.
- No declaration or use of arrays.
- No floating-point numbers.
- No string variables.

1.2 What to Submit

You should submit the following items:

- your compiler
- a file describing what changes you have to make to your scanner and parser since the previous version you turned in
- Makefile
- test programs

1.3 Java Assembler

The Java Bytecode Assembler (or simply Assembler) is a program that converts code written in "Java Assembly Language" into a valid Java .class file. It is available for non-commercial purpose. The program can be downloaded at the class home page. An online reference of Java instructions can be found at http://en.wikipedia.org/wiki/Java_bytecode_instruction_listings.

2 Generating Java Assembly Code

This section describes the four major pieces (see Section 1) of the translation of *Rust*⁻ programs into Java assembly code. This document presents methods for code generation in each piece and gives sample *Rust*⁻ along with the Java assembly code. Please note the label numbers in the examples are arbitrarily assigned. In addition, an extra piece will be added to generate code for initialization.

2.1 Initialization

A *Rust*⁻ program is translated into a Java class. An empty *Rust*⁻ program `example.rust`

```
fn main( )  
{ }
```

will be translated into the following Java assembly code

```
class example  
{  
    method public static void main(java.lang.String[])  
        max_stack 15  
        max_locals 15  
        {  
            return  
        }  
}
```

Consequently, once the file name of the source program is read, a declaration of the corresponding class name must be generated. Furthermore, a method `main` that is declared `public` and `static` must be generated for the `main` method.

2.2 Declarations for Variables and Constants

Before generating Java assembly commands for *Rust*⁻ statements, you have to allocate storage for declared variables and store values of constants.

2.2.1 Allocating Storage for Variables

Variables can be classed into two types: global and local. All variables that are declared outside functions are global, while other variables are local. All global variables must be declared before function declarations.

Global Variables

Global variables will be modeled as fields of classes in Java assembly language. Fields will be declared right after class name declaration. Each global variable `var` will be declared as a static field by the form

```
field static type var < = const_value >
```

where `type` is the type of variable `var` and `< =const_value >` is an optional initial constant value. For example,

```
let mut a = 100;
let mut b;
let mut c: Int = 200;
```

will be translated into the following declaration statements in Java assembly

```
field static integer a = 100
field static integer b
field static integer c = 200
```

Local Variables

Instance variables in *Rust*⁻ will be translated into local variables of methods in Java assembly. Unlike fields (i.e. global variables of *Rust*⁻), local variables will not be declared explicitly in Java assembly programs. Instead, local variables will be numbered and instructions to reference local variables take an integer operand indicating which variable to use. In order to number local variables, symbol tables should maintain a counter to specify “the next available number”. For example, consider the following program fragment:

```
let mut i;
let mut j;
let mut k;
```

the symbol table information will be

```
entering block, next number 0
i = 0, next number 1
j = 1, next number 2
k = 2, next number 3
leaving block, symbol table entries:
<"i", variable, integer, 0>
<"j", variable, integer, 1>
<"k", variable, integer, 2>
```

In addition, if an initialization value is given, statements must be generated to put the value onto the operand stack and then store it to the local variable. For instance, if the last statement of the above example is changed to

```
let mut k = 100;
```

a store statement must be generated as well:

```
sipush 100
istore 2           // local variable number of k is 2
```

2.2.2 Storing Constants in Symbol Table

Constant variables in *Rust*⁻ will not be transformed into fields or local variables in Java assembly. The values of constant variables will be stored in symbol tables.

2.3 Expressions and Statements

2.3.1 Expressions

An expression can be either an variable, a constant variable, an arithmetic expression, or a boolean expression.

Variables

Since operations on string variables are not considered in this project and furthermore Java Virtual Machine does not have instructions for boolean, a variable will be loaded to the operand stack by *iload* instruction if it is a local variable or *getstatic* if it is a global variable. Consider the following program fragment

```
let mut a;
...
fn foo () -> int {
    let mut b;
    ...
    = a ...
    = b ...
}
```

The translated program will contain the following Java assembly instructions

```
...
getstatic int example.a
...
iload 1  /* local variable number of b is 1 */
```

Constants

The instructions to load a constant in Java Virtual Machine is *iconst_value* or *sipush value* if the constant is a boolean or an integer, or *ldc string* if the constant is a string. Consider the following program fragment

```
let a = 10;
let b = true;
let s = "string";
...
= a ...
= b ...
print s
= 5 ...
```

The translated program will contain the following Java assembly instructions

```

sipush 10      /* = a ... */
...
iconst_1      /* = b ... */
...
ldc "string"   /* print s */
...
sipush 5       /* = 5 ... */

```

Arithmetic and Boolean Expressions

Once the compiler performs a reduction to an arithmetic expression or a boolean expression, the operands of the operation will already be on the operand stack. Therefore, only the operator will be generated. The following table lists the mapping between operators in $Rust^-$ and corresponding instructions in Java assembly language.

$Rust^-$ Operator	Integer	$Rust^-$ Operator	Boolean
+	iadd	&&	iand
-	isub		ior
*	imul	!	ixor
/	idiv		
%	irem		
- (neg)	ineg		

Boolean expressions with relational operators will be modeled by a subtraction instruction followed by a conditional jump. For instance, consider $a < b$.

```

isub /* a and b are at stack top */
iflt L1
iconst_0 /* false = 0 */
goto L2
L1: iconst_1 /* true = 1 */
L2:

```

The following table summarizes the conditional jumps for each relational operator:

$Rust^-$ relop	ifcond	$Rust^-$ relop	ifcond
<	iflt	<=	ifle
>	ifgt	=>	ifge
==	ifeq	!=	ifne

2.3.2 Statements

Assignments $id = expression;$

The right side, i.e. $expression$, will be on the operand stack when this production is reduced. As a result, the code to generate is to store the value at stack top in id . If id is a local variable, then the instruction to store the result is

```
istore 2 /* local variable number is 2 */
```

On the other hand, if *id* is global, then the instruction will be

```
putstatic type example.id
```

where *type* is the type of *id*.

PRINT Statements *print expression;*

The PRINT statement in *Rust*⁻ is modeled by invoking the *print* method in *java.io* package using the following format

```
getstatic java.io.PrintStream java.lang.System.out
... /* compute expression */
invokevirtual void java.io.PrintStream.print(java.lang.String)
```

if the type of *expression* is a string. Types *int* or *boolean* will replace *java.lang.String* if the type of *expression* is integer or logical. Similarly, a PRINTLN statement for an *expression* of the string type will be compiled to the following java assembly code:

```
getstatic java.io.PrintStream java.lang.System.out
... /* compute expression */
invokevirtual void java.io.PrintStream.println(java.lang.String)
```

2.4 If Statements and While Loops

It is fairly simple to generate code for IF and WHILE statements. Consider this if-then-else statement:

```
if (false) {
    i = 5;
}
else {
    i = 10;
}
```

The following code will be generated

```
iconst_0
ifeq Lfalse
sipush 5
istore 2 /* local variable number of i is 2 */
goto Lexit
Lfalse:
sipush 10
istore 2
Lexit:
```

For each while loop, a label is inserted before the boolean expression, and a test and a conditional jump will be performed after the boolean expression. Consider the following WHILE loop

```

n = 1;
while (n <= 10) {
    n = n + 1;
}

```

The following instructions will be generated:

```

sipush 1 /* constant 1 */
istore 1 /* local variable number of n is 1 */
Lbegin:
  iload 1
  sipush 10
  isub
  ifle Ltrue
  iconst_0
  goto Lfalse
Ltrue:
  iconst_1
Lfalse:
  ifeq Lexit
  iload 1 /* local variable number of n is 1 */
  sipush 1
  iadd
  istore 1
  goto Lbegin
Lexit:

```

2.5 Function Declaration and Invocation

Functions in *Rust*[−] will be modeled by static methods in Java assembly language.

2.5.1 Function Declaration

If n arguments are passed to a static Java method, they are received in the local variables numbered 0 through $n - 1$. The arguments are received in the order they are passed. For example

```

fn add(a:int, b:int) -> int {
    return a+b;
}

```

will be compiled to

```

method public static int add(int, int)
max_stack 15
max_locals 15
{
    iload 0
    iload 1

```

```

        iadd
        ireturn
    }

```

A procedure is declared without a return value. Hence the type of its corresponding method will be **void** and the return instruction will be **return**.

2.5.2 Function Invocation

To invoke a static method, the instruction *invokestatic* will be used. The following method invocation

```
= add(a, 10) ...
```

will be compiled into

```

...
iload 1  /* local variable number of a is 1 */
sipush 10 /* constant 10 */
invokestatic int example.add(int, int)
...

```

where the first `int` is the return type of the method and the second and last `int` are the types of formal parameters.

3 Implementation Notes

3.1 Local Variable Numbers

Formal parameters of a method are numbered starting from 0. Local variables in the method are placed right after the formal parameters of the method. For example, if a method has n formal parameters, then the parameters are numbered from 0 to $n - 1$, while the first local variable will be numbered n .

3.2 Java Virtual Machine

Once an *Rust*[−] program is compiled, the resulted Java assembly code can then be transformed into Java bytecode using the Java assembler *javaa*. The output of *javaa* will be a class file of the generated bytecode, which can be executed on the Java Virtual Machine. For example, if the generated class is *example.class*, then type the following command to run the bytecode

```
% java example
```

The structure of Java Virtual Machine is described in the book “*Java Virtual Machine Specification*”, which can be found at <http://java.sun.com/docs/books/vmspec/index.html>.

Example

Source *Rust*™ program *example.rust*:

```
/*
 * Example with Functions
 */

// variables
let mut c;
let mut a = 5;

// function declaration
fn add(a:int, b:int) -> int {
    return a+b;
}

// main function
fn main() {
    c = add(a, 10);
    if (c > 10) {
        print -c;
    }
    else {
        print c;
    }
    println "Hello World";
}
```

Generated Java assembly program:

```
/*-----*/
/*          Java Assembly Code          */
/*-----*/

class example
{
/* 4:  */
/* 5: // variables */
    field static int c
/* 6: let mut c; */
    field static int a = 5
/* 7: let mut a = 5; */
/* 8:  */
/* 9: // fuction declaration */
    method public static int add(int, int)
        max_stack 15
        max_locals 15
    {
/* 10: fn add(a:int, b:int) -> int { */
        iload 0
        iload 1
        iadd
        ireturn
/* 11:    return a+ b; */
    }
/* 12: } */
/* 13: */
/* 14: // main function */
    method public static void main(java.lang.String[])
        max_stack 15
        max_locals 15
    {
/* 15: fn main( ) { */
        getstatic int example.a
        sipush 10
        invokestatic int example.add(int, int)
        putstatic int example.c
/* 16:    c = add(a, 10); */
        getstatic int example.c
        sipush 10
        isub
        ifgt L0
        iconst_0
        goto L1
    }
}
```

```

L0:
    iconst_1
L1:
    ifeq L2
/* 17:    if (c > 10) { */
    getstatic java.io.PrintStream java.lang.System.out
    getstatic int example.c
    ineg
    invokevirtual void java.io.PrintStream.print(int)
/* 18:        print -c; */
/* 19:    }
    goto L3
L2:
/* 20:    else { */
    getstatic java.io.PrintStream java.lang.System.out
    getstatic int example.c
    invokevirtual void java.io.PrintStream.print(int)
/* 21:        print c; */
/* 22:    } */
L3:
    getstatic java.io.PrintStream java.lang.System.out
    ldc "Hello World"
    invokevirtual void java.io.PrintStream.println(java.lang.String)
/* 23:    println "Hello World"; */
    return
    }
/* 24: } */
}

```